
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2013-03_D15wk18’.

3. Correction to TCs 8.2.4.16.x
Change 1 – Correction to function ‘f_EUTRA_508RRC_AddModRel_Scell_MIMO’
	Function name
	f_EUTRA_508RRC_AddModRel_Scell_MIMO ()

	Reason for change
	Dedicated antenna config of the SCell is timely not in sync with the other SCell configurations.

	Summary of change
	Move the antenna config part to the SCell configurations thus using proper timing info.

	TTCN module
	\LTE-A\Common\EUTRA_CommonFunctions_CA.ttcn

	MCC160 Comment
	


Before change

...
  function f_EUTRA_508RRC_AddModRel_Scell_MIMO(EUTRA_CellId_Type p_PCellId,
                                               EUTRA_CellId_Type p_SCellId,
                                               template (value) DL_DCCH_Message p_RRC_ConnRecongMsg,
                                               template (value) Scell_Capability_Type p_Scell_Capability := UL_DL,
                                               template (omit) MeasGapConfig p_MeasGapConfig := omit,
                                               template (omit)  DciUlInfo_Type p_DciUlInfo_SCell := cs_DciInfo_CcchDcchDtchUL_Def,
                                               template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SCell := cs_UL_GrantConfig_Def)
    runs on EUTRA_PTC
  {
    var boolean v_MeasGapConfigIsPresent := isvalue(p_MeasGapConfig);
    var DrxCtrl_Type v_DrxCtrl;
    var RRCConnectionReconfiguration_r8_IEs v_RRCConnectionReconfigurationIEs := valueof(p_RRC_ConnRecongMsg.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8);
    var SCellIndex_r10 v_SCellIndex := 1;
    var MAC_MainConfig_ScellDeactivationTimer_Type v_DeactTimer := rf32;
    var boolean v_NoUL_SCell ; //@sic R5s140083 sic@
    var template (omit) STAG_Id_r11 v_STAG_Id := omit ; //@sic new CA-Enh test development sic@
    var SubFrameTiming_Type v_TimingInfo;
    var SubFrameTiming_Type v_TimingInfo2;
    // Step 1: Configure PCell as Pcell or normal cell
    f_EUTRA_SS_CommonCellConfig(p_PCellId, cas_PCellConfig_REQ(p_PCellId, cs_TimingInfo_Now, {p_SCellId}));
    // Configure Scell for TM change
    if (ispresent(v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.physicalConfigDedicatedSCell_r10.nonUL_Configuration_r10.antennaInfo_r10)) {
      f_EUTRA_AntennaInfoR10_Config(p_SCellId, cs_TimingInfo_Now, v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.physicalConfigDedicatedSCell_r10.nonUL_Configuration_r10.antennaInfo_r10);
    }
    if (isvalue(v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.mac_MainConfig.explicitValue.mac_MainConfig_v1020.sCellDeactivationTimer_r10)) {
      v_DeactTimer := v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.mac_MainConfig.explicitValue.mac_MainConfig_v1020.sCellDeactivationTimer_r10;
    }
    v_NoUL_SCell := f_EUTRA_CellInfo_GetNoUL_Frequency(p_SCellId);
    v_SCellIndex := v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].sCellIndex_r10;
    if (not v_NoUL_SCell) {
      // Step 2: Configure C-RNTI based contention resolution in S-cell
      f_EUTRA_SS_ConfigRachProcedure(p_SCellId, tsc_C_RNTI_Def, cs_RachProcedureConfig_CRNTI(f_EUTRA_CellInfo_GetRAR_TA(p_SCellId),
                                                                                             f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId)));
    }
    // Save new C-RNTI in cell configuration for p_SCellId
    f_EUTRA_CellInfo_SetC_RNTI(p_SCellId, tsc_C_RNTI_Def);
    // Step 3: configure DRX and MeasGap at the S cell (if necessary)
    if (f_EUTRA_CellInfo_DrxIsConfigured(p_PCellId)) {
      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_PCellId);
      f_EUTRA_SS_DrxCtrlConfig(p_SCellId, v_DrxCtrl);
    }
    if (v_MeasGapConfigIsPresent) {
      f_EUTRA_SS_MeasGapCtrlConfig(p_SCellId, cs_MeasGapCtrl_Config(valueof(p_MeasGapConfig)));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present
    }
    if (ispresent (v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.mac_MainConfigSCell_r11)) {
      v_STAG_Id := v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.mac_MainConfigSCell_r11.stag_Id_r11;
    }
    v_TimingInfo := f_EUTRA_GetNextSendOccasion(p_PCellId);
    v_TimingInfo2 := f_EUTRA_TimingInfoAdd(v_TimingInfo,5); // 5 ms later
    // Step 6: Transmit RRCConnectionReconfiguration/HO command to UE
    SRB.send(cas_SRB1_RrcPdu_REQ(p_PCellId, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), p_RRC_ConnRecongMsg));
    // Configure Pcell for TM change
    if (ischosen (v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.physicalConfigDedicated.antennaInfo_r10.explicitValue_r10)) {
      f_EUTRA_AntennaInfoR10_Config(p_PCellId,
                                    cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number),
                                    v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.physicalConfigDedicated.antennaInfo_r10.explicitValue_r10);
    }
    // Step 4: Configure SCell as Scell
    f_EUTRA_SS_CommonCellConfig(p_SCellId, cas_SCellConfig_REQ(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), cs_SCellConfig(p_PCellId, v_SCellIndex, p_Scell_Capability, v_DeactTimer,-,v_STAG_Id)));
    if (not v_NoUL_SCell) {
      // Step 5: configure Stop UL grant configuration on S cell
      f_EUTRA_SS_CommonCellConfig(p_SCellId, cas_ULGrantAllocation_REQ(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), p_DciUlInfo_SCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_SCell));
    }
    // Step 7: Receive RRCConnectionReconfiguration Complete
    SRB.receive(car_SRB1_RrcPdu_IND(p_PCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
    // No RACH configuration as Already configured for C-RNTI based contention
  }
...
After change

...
  function f_EUTRA_508RRC_AddModRel_Scell_MIMO(EUTRA_CellId_Type p_PCellId,
                                               EUTRA_CellId_Type p_SCellId,
                                               template (value) DL_DCCH_Message p_RRC_ConnRecongMsg,
                                               template (value) Scell_Capability_Type p_Scell_Capability := UL_DL,
                                               template (omit) MeasGapConfig p_MeasGapConfig := omit,
                                               template (omit)  DciUlInfo_Type p_DciUlInfo_SCell := cs_DciInfo_CcchDcchDtchUL_Def,
                                               template (value) UL_GrantConfig_Type  p_UL_GrantConfig_SCell := cs_UL_GrantConfig_Def)
    runs on EUTRA_PTC
  {
    var boolean v_MeasGapConfigIsPresent := isvalue(p_MeasGapConfig);
    var DrxCtrl_Type v_DrxCtrl;
    var RRCConnectionReconfiguration_r8_IEs v_RRCConnectionReconfigurationIEs := valueof(p_RRC_ConnRecongMsg.message_.c1.rrcConnectionReconfiguration.criticalExtensions.c1.rrcConnectionReconfiguration_r8);
    var SCellIndex_r10 v_SCellIndex := 1;
    var MAC_MainConfig_ScellDeactivationTimer_Type v_DeactTimer := rf32;
    var boolean v_NoUL_SCell ; //@sic R5s140083 sic@
    var template (omit) STAG_Id_r11 v_STAG_Id := omit ; //@sic new CA-Enh test development sic@
    var SubFrameTiming_Type v_TimingInfo;
    var SubFrameTiming_Type v_TimingInfo2;
    // Step 1: Configure PCell as Pcell or normal cell
    f_EUTRA_SS_CommonCellConfig(p_PCellId, cas_PCellConfig_REQ(p_PCellId, cs_TimingInfo_Now, {p_SCellId}));
    if (isvalue(v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.mac_MainConfig.explicitValue.mac_MainConfig_v1020.sCellDeactivationTimer_r10)) {
      v_DeactTimer := v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.mac_MainConfig.explicitValue.mac_MainConfig_v1020.sCellDeactivationTimer_r10;
    }
    v_NoUL_SCell := f_EUTRA_CellInfo_GetNoUL_Frequency(p_SCellId);
    v_SCellIndex := v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].sCellIndex_r10;
    if (not v_NoUL_SCell) {
      // Step 2: Configure C-RNTI based contention resolution in S-cell
      f_EUTRA_SS_ConfigRachProcedure(p_SCellId, tsc_C_RNTI_Def, cs_RachProcedureConfig_CRNTI(f_EUTRA_CellInfo_GetRAR_TA(p_SCellId),
                                                                                             f_EUTRA_CellInfo_GetDL_ChBandwidth(p_SCellId)));
    }
    // Save new C-RNTI in cell configuration for p_SCellId
    f_EUTRA_CellInfo_SetC_RNTI(p_SCellId, tsc_C_RNTI_Def);
    // Step 3: configure DRX and MeasGap at the S cell (if necessary)
    if (f_EUTRA_CellInfo_DrxIsConfigured(p_PCellId)) {
      v_DrxCtrl := f_EUTRA_CellInfo_GetDrxCtrl(p_PCellId);
      f_EUTRA_SS_DrxCtrlConfig(p_SCellId, v_DrxCtrl);
    }
    if (v_MeasGapConfigIsPresent) {
      f_EUTRA_SS_MeasGapCtrlConfig(p_SCellId, cs_MeasGapCtrl_Config(valueof(p_MeasGapConfig)));   // valueof to avoid compiler warnings: ok as v_MeasConfig.measGapConfig is checked to be present
    }
    if (ispresent (v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.mac_MainConfigSCell_r11)) {
      v_STAG_Id := v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.mac_MainConfigSCell_r11.stag_Id_r11;
    }
    v_TimingInfo := f_EUTRA_GetNextSendOccasion(p_PCellId);
    v_TimingInfo2 := f_EUTRA_TimingInfoAdd(v_TimingInfo,5); // 5 ms later
    // Step 6: Transmit RRCConnectionReconfiguration/HO command to UE
    SRB.send(cas_SRB1_RrcPdu_REQ(p_PCellId, cs_TimingInfo(v_TimingInfo.SFN.Number, v_TimingInfo.Subframe.Number), p_RRC_ConnRecongMsg));
    // Configure Pcell for TM change
    if (ischosen (v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.physicalConfigDedicated.antennaInfo_r10.explicitValue_r10)) {
      f_EUTRA_AntennaInfoR10_Config(p_PCellId,
                                    cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number),
                                    v_RRCConnectionReconfigurationIEs.radioResourceConfigDedicated.physicalConfigDedicated.antennaInfo_r10.explicitValue_r10);
    }
    // Step 4: Configure SCell as Scell
    f_EUTRA_SS_CommonCellConfig(p_SCellId, cas_SCellConfig_REQ(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), cs_SCellConfig(p_PCellId, v_SCellIndex, p_Scell_Capability, v_DeactTimer,-,v_STAG_Id)));
    if (not v_NoUL_SCell) {
    // Step 5: configure Stop UL grant configuration on S cell
    f_EUTRA_SS_CommonCellConfig(p_SCellId, cas_ULGrantAllocation_REQ(p_SCellId, cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number), p_DciUlInfo_SCell, cs_PUCCH_Synch_None, p_UL_GrantConfig_SCell));
    }
    // Configure Scell for TM change
    if (ispresent(v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.physicalConfigDedicatedSCell_r10.nonUL_Configuration_r10.antennaInfo_r10)) {
      f_EUTRA_AntennaInfoR10_Config(p_SCellId,
                                    cs_TimingInfo(v_TimingInfo2.SFN.Number, v_TimingInfo2.Subframe.Number),
                                    valueof(v_RRCConnectionReconfigurationIEs.nonCriticalExtension.nonCriticalExtension.nonCriticalExtension.sCellToAddModList_r10[0].radioResourceConfigDedicatedSCell_r10.physicalConfigDedicatedSCell_r10.nonUL_Configuration_r10.antennaInfo_r10));
    }
    // Step 7: Receive RRCConnectionReconfiguration Complete
    SRB.receive(car_SRB1_RrcPdu_IND(p_PCellId, cr_508_RRCConnectionReconfigurationComplete(tsc_RRC_TI_Def)));
    // No RACH configuration as Already configured for C-RNTI based contention
  }
... 
